Mathematica入門

現今中學數學課程，幾乎停留在紙筆的抽象研究，難以實際瞭解或進行更複雜的操作。鑑於此，藉由 Mathematica 高速運算特色的數學工具，來進一步對高中數學加以驗證，也提供大家進行研究專題時的另一種絕佳的輔助途徑。現在就讓我們一起體驗Mathematica 的奧妙吧！

軟體簡介

Mathematica 是一套整合「數字」、「符號運算」、「排版」的數學工具軟體，提供了全球超過百萬的研究人員，工程師，物理學家，分析師以及其他技術專業人員容易使用的頂級科學運算環境。目前已在學術界、電機、機械、化學、土木、資訊工程、財務金融、醫學、物理、統計、教育出版、OEM (Original Equipment Manufacturer) 等廣泛使用。（本段落節錄自 http://www.sciformosa.com.tw/products/mathematica.asp）

本講義便是以 Mathematica 軟體排版完成，完全沒有藉由其他任何軟體進行輔助。

Mathematica 的特色

◎ 將時間花在問題上而非計算

◎ 內建龐大的數學知識庫

◎ 輸出結果呈現視覺化效果

◎ 呈現精確的數值運算結果

◎ 享受高速的電腦代數運算

◎ 自動選取最佳化

◎ 與 txt、html、word 格式相容

◎ 完全可設計的符號表示

◎ 縮短研發時程

第零章、認識操作環境

1.開啓Mathematica軟體

開始 → 程式集 → Mathematica5 → 執行Mathematica5

2.常用設定與準備工作

(1) 開啓常用面版：

File → Palettes → 2 Algebraic Manipulation （代數常用函數化簡工具）

File → Palettes → 4 Basic Input（常用函數式子）

File → Palettes → 5 Basic Typsetting （常用基本符號）

(2) 設定執行模式

標準輸入模式：Cell → Default Input Format Style → StandardForm

傳統輸出模式：Cell → Default Onput Format Style → TranditionalForm

儲存檔案（副檔名爲.nb）：File → Save → 檔名.nb

「xxx.nb」是 Mathematica 的檔案名稱

第一章、基本四則運算

1.執行的指令為：[Shift]+[Enter]

範例1 計算 1+2+3+4 之值

於工作視窗輸入「1+2+3+4」，接著按下[Shift]+[Enter]即可執行

In[3]:= **1 + 2 + 3 + 4**

Out[3]= 10

編輯畫面中左方出現的標號「In[1]:=」 代表第 1 次輸入的內容，
「Out[1]:=」代表第 1 次輸出的內容。
第一次執行數值計算時，會自動將核心檔案讀到記憶體內，因此會花上較多的時間等待，第二次之後所執行的速度便會相當快。另外，若執行時等待太久，可以按下 [Alt]+[,] 或 [Alt]+[.] 來立即強迫中斷或結束運算。

2. 有理數的四則運算

範例2 利用常用面版計算 $\frac{12\times34\times(56+78)}{9}$ 之值。

「空格」代表乘法運算。結果出現分數，而不會像計算機給近似值。

$$\frac{12\ 34\ (56+78)}{9}$$

$$\frac{18224}{3}$$

範例3 引用範例 2 的結果，計算 $\frac{12\times34\times(56+78)}{9}+\frac{10}{11}$ 之值

「%2」代表第2次的輸出結果，「%n」代表第 n 次的輸出結果

$$\%+\frac{10}{11}$$

$$\frac{200494}{33}$$

範例4 利用 N 將 $\frac{1}{7}$ 以小數型態表示

「//N」或「N[ ]」即可表示出近似數值，預設值為六位精準。

$\frac{1}{7}$ // N

0.142857

N[$\frac{1}{7}$]

0.142857

範例5 計算 $\frac{1}{7}$ 的值（精準50位）

$N[\frac{1}{7}, n]$ 表示精準 n 位，其中 n>16 。

N[$\frac{1}{7}$, 50]

0.14285714285714285714285714285714285714285714285714

練習6 計算 2^{100}

馬上輸出實際值，而非科學記號！

2^{100}

1267650600228229401496703205376

3. 無理數的四則運算

範例7 計算 $\sqrt{3}+\sqrt{27}+\sqrt{720}$

自動輸出化簡結果

$$\sqrt{3}+\sqrt{27}+\sqrt{720}$$

$$4\sqrt{3}+12\sqrt{5}$$

$$\mathbf{4\sqrt{3}+12\sqrt{5}}$$

範例8 計算 $\sqrt{10-2\sqrt{21}}$

發現輸出結果還使以 $\sqrt{10-2\sqrt{21}}$ 呈現

$$\sqrt{10-2\sqrt{21}}$$

$$\sqrt{10-2\sqrt{21}}$$

範例9 利用輸入面版 FullSimplify 函數，來化簡雙重根號 $\sqrt{10-2\sqrt{21}}$

發現輸出結果將去除雙重根號

$$\sqrt{10-2\sqrt{21}} \;\texttt{// FullSimplify}$$

$$-\sqrt{3}+\sqrt{7}$$

$$\texttt{FullSimplify}\left[\sqrt{10-2\sqrt{21}}\right]$$

$$-\sqrt{3}+\sqrt{7}$$

練習10 利用輸入面版 FullSimplify 函數，來化簡 $\left(11-6\sqrt{2}\right)^{-\frac{1}{2}}\left(3+\sqrt{2}\right)^{-1}$

快速將化簡結果呈現！

$$\left(11-6\sqrt{2}\right)^{-\frac{1}{2}}\left(3+\sqrt{2}\right)^{-1} \text{ // FullSimplify}$$

$$\frac{1}{7}$$

4. 常用常數

常數名稱	鍵盤輸入	面版輸入	快速按鍵	數值
圓周率	Pi	π	< esc > p < esc >	3.14159 …
尤拉常數	E	e	< esc > ee < esc >	2.71828 …
角度	Degree	°	< esc > deg < esc >	
虛數	I	i	< esc > ii < esc >	
無限大	Infinity	∞	< esc > inf < esc >	

範例11 半徑為 $\frac{2}{3}$ 的圓面積為何？

試利用面版或快速按鍵完成計算

$$\left(\frac{2}{3}\right)^{2}\pi$$

$$\frac{4\pi}{9}$$

範例12 求圓周率 π 精準100位

利用N[π,1000]即可。就算想瞭解十萬位，亦可在五秒內得知！

```
In[4]:= N[π, 100]
Out[4]= 3.141592653589793238462643383279502884197169399375105820974944592307816406286208998628034825342117068
```

範例13 求尤拉常數（自然指數數）e 精準100位

In[5]:= `N[e, 100]`

Out[5]= 2.718281828459045235360287471352662497757247093699959574966967627724076630353547594571382178525166427

5. 複數的四則運算

範例14 計算 $\sqrt{-1}$

$\sqrt{-1}$

i

練習15 計算 $\left(\frac{1+i}{\sqrt{2}}\right)^{100}$

$\left(\frac{1+i}{\sqrt{2}}\right)^{100}$

-1

練習16 計算 $(1+2i)(3-4i)^5$

將自動化簡成 $a+bi$ 的形式

$(1+2i)(3-4i)^5$

$-6469+2642i$

範例17 $x = 1 + \sqrt{3}i$ ，試求x的實部、虛部、共軛複數、大小(與原點的距離)、主幅角

💡 實部： Re[x]　　虛部：Im[x]　　共軛複數：Conjugate[x]　大小（與原點距離）：Abs[x]　　主幅角：Arg[x]

x = 1 + $\sqrt{3}$ ⅈ

$1 + i\sqrt{3}$

Re[x]

1

Im[x]

$\sqrt{3}$

Conjugate[x]

$1 - i\sqrt{3}$

Abs[x]

2

Arg[x]

$\frac{\pi}{3}$

Abs[x]

2

範例18　將 $(-1)^{1/3}$ 表式成 a + bi 的型態。

💡 利用 ComplexExpand 函數強迫以 a+bi 型態展現。

In[24]:= **$(-1)^{1/3}$ // ComplexExpand**

Out[24]= $\frac{1}{2} + \frac{i\sqrt{3}}{2}$

6. 常用函數

內建函數的第一個字母均為大寫，函數以中括號表示。

函數名稱	使用方法
三角函數	Sin[x] 、Cos[x] 、Tan[x] 、Cot[x] 、Sec[x] 、Csc[x]
反三角函數	ArcSin[x] 、ArcCos[x] 、ArcTan[x] 、ArcCot[x] 、ArcSec[x] 、ArcCsc[x]
根號	Sqrt[x]（即 $\sqrt{2}$）
對數函數	Log[n, x]（以n為底數，x為真數之值，即 $\log_n x$）
自然對數函數	Log[x]（以 e 為底數，x為真數之值，即。注意並不是以10為底）
絕對值	Abs[x]（即計算 $\lvert x \rvert$）
四捨五入至整數位	Round[x]
下高斯	Floor[x]（即小於或等於x的最大整數）
上高斯	Ceiling[x]（即大於或等於x的最小整數）
餘數（模）	Mod[a, b]（求a除以b的餘數，其中a, b均為整數）
階乘	n!（求 $n\times(n-1)\times\ldots\times2\times1$ 之值）
亂數	Random[]（求 (0, 1) 區間的亂數）
刪除之前的指定	Clear[x]

範例18 計算 100!

100!

9332621544394415268169923885626670049071596826438162146859296389521759999322991560894146397615651828625369792082722375825118521091686400000000000000000000000000

第二章、符號運算與多項式

項目	語法
展開	Expand[多項式]
徹底展開	ExpandAll[多項式]
因式分解	Factor[多項式]
通分後合併	Together[多項式]
部分分式	Apart[多項式]
消去共同因子	Cancel[多項式]
化簡多項式	Simplify[多項式]
取出分式的分母	Denominator[分式]
取出分式的分子	Numerator[分式]
取出係數	Coefficient[多項式, 指定項]
取出多項式的最高次數	Exponent[多項式, 指定項]
取出第n項	Part[多項式, n]

範例1 展開多項式 $(x+y)^4$

利用 Expand 函數

In[6]:= `Expand[(x + y)^4]`

Out[6]= $x^4 + 4\,y\,x^3 + 6\,y^2\,x^2 + 4\,y^3\,x + y^4$

In[7]:= `(x + y)^4 // Expand`

Out[7]= $x^4 + 4\,y\,x^3 + 6\,y^2\,x^2 + 4\,y^3\,x + y^4$

範例2 將上式展開的結果進行因式分解

利用 % 符號，以及 Factor 函數

In[8]:= `% // Factor`

Out[8]= $(x+y)^4$

範例3 因式分解 $2x^2+xy-3x-6y^2+8y-2$

💡 注意 x y 中間必須要有空格代表相乘，否則將視爲一個變數叫做 xy

In[9]:= $2x^2+xy-3x-6y^2+8y-2$ // Factor

Out[9]= $(2x-3y+1)(x+2y-2)$

範例4 因式分解 $x^3+y^3+z^3-3\,\mathrm{xyz}$

💡 注意 x y z 中間必須要用空格，代表相乘

In[10]:= $x^3+y^3+z^3-3\,x\,y\,z$ // Factor

Out[10]= $(x+y+z)(x^2-yx-zx+y^2+z^2-yz)$

範例5 化簡 $\frac{(3x+2)^2}{(6x+4)^2}$

💡 利用 Simply 函數進行化簡，若發現不能化簡，則可嘗試用 FullSimplify 化簡之

In[11]:= $\frac{(3x+2)^2}{(6x+4)^2}$ // Simplify

Out[11]= $\frac{1}{4}$

第三章、三角函數

範例1　計算 210° 的六個三角函數值。

注意函數的第一個字母要大寫。

In[57]:=
```
{Sin[210 °], Cos[210 °], Tan[210 °],
 Cot[210 °], Sec[210 °], Csc[210 °]}
```

Out[57]= $\left\{-\frac{1}{2}, -\frac{\sqrt{3}}{2}, \frac{1}{\sqrt{3}}, \sqrt{3}, -\frac{2}{\sqrt{3}}, -2\right\}$

練習2　計算 15° 的六個三角函數值.

In[66]:=
```
{Sin[15 °], Cos[15 °], Tan[15 °], Cot[15 °], Sec[15 °], Csc[15 °]}
```

Out[66]= $\left\{\frac{-1+\sqrt{3}}{2\sqrt{2}}, \frac{1+\sqrt{3}}{2\sqrt{2}}, 2-\sqrt{3}, 2+\sqrt{3}, \sqrt{2}\left(-1+\sqrt{3}\right), \sqrt{2}\left(1+\sqrt{3}\right)\right\}$

範例3　計算 $\sin\frac{\pi}{8}$

利用 FunctionExpand 強迫展開三角函數

In[63]:= $\texttt{Sin}\left[\frac{\pi}{8}\right]$ `// FunctionExpand`

Out[63]= $\frac{\sqrt{2-\sqrt{2}}}{2}$

練習4　計算 $\sin\frac{\pi}{24}$

In[67]:= $\texttt{Sin}\left[\frac{\pi}{24}\right]$ `// FunctionExpand`

Out[67]= $-\frac{1}{4}\sqrt{3\left(2-\sqrt{2}\right)}+\frac{\sqrt{2+\sqrt{2}}}{4}$

範例5　計算 $\sin 270^\circ+\cos 180^\circ+\cot 90^\circ$

```
In[72]:= Sin[270 °] + Cos[180 °] + Cot[90 °]
Out[72]= -2
```

範例6　計算 $\cos 1770^\circ \tan 1110^\circ+\sin(-1560^\circ) \cot 510^\circ$

```
In[73]:= Cos[1770 °] Tan[1110 °] + Sin[-1560 °] Cot[510 °]
Out[73]= 2
```

練習7　計算 $\sin 23^\circ \cos 112^\circ-\sin 292^\circ \sin 67^\circ$

利用 FullSimplify 展開

```
In[75]:= Sin[23 °] Cos[112 °] - Sin[292 °] Sin[67 °] // FullSimplify
```

Out[75]= $\frac{1}{\sqrt{2}}$

範例8　試比較 $\sin 115^\circ$ 、 $\cos 732^\circ$ 、 $\tan 332^\circ$ 的大小。

```
In[26]:= {Sin[115 °], Cos[732 °], Tan[332 °]} // N
Out[26]= {0.906308, 0.978148, -0.531709}
```

範例9　比較 sin1, sin2, sin3, sin4, sin5,sin6 的大小。

In[29]:=
```
{Sin[1], Sin[2], Sin[3], Sin[4], Sin[5], Sin[6]} // N
```
Out[29]= {0.841471, 0.909297, 0.14112, −0.756802, −0.958924, −0.279415}

練習10　比較 cos(-2), cos(-1), cos(0), cos1, cos2, cos3 的大小。

In[31]:=
```
{Cos[-2], Cos[-1], Cos[0], Cos[1], Cos[2], Cos[3]} // N
```
Out[31]= {−0.416147, 0.540302, 1., 0.540302, −0.416147, −0.989992}

練習11　平面上的點 (sin100, tan50) 在第幾象限？

In[32]:=
```
{Sin[100], Tan[50]} // N
```
Out[32]= {−0.506366, −0.271901}

練習12　計算化簡 $\cos\frac{\pi}{7}\cos\frac{2\pi}{7}\cos\frac{4\pi}{7}$

先利用 TrigExpand 函數將三角函數角度統一，再利用 FullSimplify 化簡。

In[56]:= $\text{Cos}\left[\frac{\pi}{7}\right]\text{Cos}\left[\frac{2\pi}{7}\right]\text{Cos}\left[\frac{4\pi}{7}\right]$ // TrigExpand // FullSimplify

Out[56]= $-\frac{1}{8}$

練習13　計算化簡 $\sin^2 x + \cos^2 x$

In[48]:= $(\text{Sin}[x])^2 + (\text{Cos}[x])^2$ // Simplify

Out[48]= 1

練習14　利用 Simplify 或 TrigExpand 來驗證以下各式：

(1) $\tan^2 x - \sin^2 x = \tan^2 x \sin^2 x$

(2) $\sin(\alpha+\beta)=\sin\alpha\cos\beta+\cos\alpha\sin\beta$

(3) $\frac{\sin\theta+2\sin\theta\cos\theta}{1+\cos\theta+\cos^2\theta-\sin^2\theta} = \tan\theta$

```
In[57]:= (Tan[x])^2 - (Sin[x])^2 // Simplify
```

Out[57]= $\sin^2(x)\tan^2(x)$

```
In[60]:= Sin[α + β] // TrigExpand
```

Out[60]= $\cos(\beta)\sin(\alpha)+\cos(\alpha)\sin(\beta)$

```
In[62]:= (Sin[θ] + 2 Sin[θ] Cos[θ])/(1 + Cos[θ] + (Cos[θ])^2 - (Sin[θ])^2) // Simplify
```

Out[62]= $\tan(\theta)$

第四章、數列與級數

範例1 數列 $\{\frac{n}{n+1}\}$ 是收斂或爲發散數列？

利用 Limit[$\frac{n}{n+1}$,n→∞]

```
In[155]:= Limit[n/(n + 1), n → ∞]
```

Out[155]= 1

練習2 試計算下列各值：

$\lim_{x\to\infty}\left(\frac{4x^2-1}{4x^2+8x+3}\right) = ?$

$\lim_{x\to 0}\left(\frac{\sqrt{1+x}-\sqrt{1-x}}{x}\right) = ?$

$\lim_{x\to 4}\left(\frac{x^2-16}{2-\sqrt{x}}\right) = ?$

$\lim_{x\to 0}\left(\frac{(1+x)^2-1}{x}\right) = ?$

$\lim_{x\to16}\left(\frac{x^{\frac{3}{4}}-8}{x-16}\right)=?$

$\lim_{x\to0}\left(\frac{10^{x}-1}{x}\right)=?$

$\lim_{x\to-\infty}\left(\frac{3^{x}-3^{-x}}{3^{x}+3^{-x}}\right)=?$

$\lim_{x\to\infty}\left(1+\frac{1}{x}\right)^{x}=?$

$\lim_{x\to\infty}\left(1+\frac{z}{x}\right)^{x}=?$

$\lim_{x\to0}\left(\frac{1-\cos(2x)}{x}\right)=?$

$\lim_{x\to0}\left(\frac{\sin x}{x}\right)^{\frac{1}{x^{2}}}=?$

注意語法與函數的指令

In[12]:= $\text{Limit}\left[\frac{4x^2-1}{4x^2+8x+3},\ x\to\infty\right]$

Out[12]= 1

In[13]:= $\text{Limit}\left[\frac{\sqrt{1+x}-\sqrt{1-x}}{x},\ x\to0\right]$

Out[13]= 1

In[14]:= $\text{Limit}\left[\frac{x^2-16}{2-\sqrt{x}},\ x\to4\right]$

Out[14]= -32

In[18]:= $\text{Limit}\left[\frac{(1+x)^2-1}{x},\ x\to0\right]$

Out[18]= 2

In[19]:= $\text{Limit}\left[\frac{x^{\frac{3}{4}}-8}{x-16},\ x\to16\right]$

Out[19]= $\frac{3}{8}$

In[20]:= $\text{Limit}\left[\frac{10^x-1}{x},\ x\to0\right]$

Out[20]= $\log(10)$

In[21]:= $\text{Limit}\left[\frac{3^x-3^{-x}}{3^x+3^{-x}},\ x\to-\infty\right]$

Out[21]= -1

In[31]:= $\text{Limit}\left[\left(1+\frac{1}{x}\right)^x,\ x\to\infty\right]$

Out[31]= e

In[30]:= $\texttt{Limit}\left[\left(1+\frac{z}{x}\right)^{x},\ x\to\infty\right]$

Out[30]= e^{z}

In[29]:= $\texttt{Limit}\left[\frac{1-\texttt{Cos}[2\,x]}{x},\ x\to 0\right]$

Out[29]= 0

In[28]:= $\texttt{Limit}\left[\left(\frac{\texttt{Sin}[x]}{x}\right)^{\frac{1}{x^2}},\ x\to 0\right]$

Out[28]= $\frac{1}{\sqrt[6]{e}}$

練習3 化簡 $\sum_{k=1}^{n} k$、$\sum_{k=1}^{n} k^2$、$\sum_{k=1}^{n} k^3$

利用面版完成計算

In[32]:= $\sum_{k=1}^{n} k$

$\sum_{k=1}^{n} k^2$

$\sum_{k=1}^{n} k^3$

Out[32]= $\frac{1}{2}n\,(n+1)$

Out[33]= $\frac{1}{6}n\,(n+1)\,(2\,n+1)$

Out[34]= $\frac{1}{4}n^2\,(n+1)^2$

練習4 化簡 $\sum_{k=1}^{n}\frac{1}{k(k+1)}$、$\sum_{k=1}^{n}\frac{1}{k(k+1)(k+2)}$、$\sum_{k=1}^{\infty}\frac{1}{2^k}$

In[39]:= $\sum_{k=1}^{n} \frac{1}{k\ (k+1)}$

$\sum_{k=1}^{n} \frac{1}{k\ (k+1)\ (k+2)}$

$\sum_{k=1}^{\infty} \frac{1}{2^k}$

Out[39]= $\frac{n}{n+1}$

Out[40]= $\frac{n\,(n+3)}{4\,(n+1)\,(n+2)}$

Out[41]= 1

範例5 列出1到20連續的整數

利用 Range 函數

In[42]:= **Range[20]**

Out[42]= {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20}

範例6 列出3到20連續的整數

In[43]:= **Range[3, 20]**

Out[43]= {3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20}

範例7 在3到20連續的整數中，每4個列一次

Range[3, 20, 4]

{3, 7, 11, 15, 19}

範例8 列出 {n}數列的前 50 項

💡 利用 Table 函數，並指定 n 的起始值與結束值。

In[44]:= `Table[n, {n, 1, 50}]`

Out[44]= {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, 26, 27, 28, 29, 30, 31, 32, 33, 34, 35, 36, 37, 38, 39, 40, 41, 42, 43, 44, 45, 46, 47, 48, 49, 50}

範例9 列出 $\{5^n\}$數列的前10項

💡 利用 Table 函數，並指定 n 的起始值與結束值。

In[47]:= `Table[5^n, {n, 1, 10}]`

Out[47]= {5, 25, 125, 625, 3125, 15625, 78125, 390625, 1953125, 9765625}

第五章、指數與對數

範例1 化簡 $\left(51-14\sqrt{2}\right)^{\frac{1}{2}}-625^{\frac{1}{4}}+16^{\frac{1}{8}}$

💡 利用 Simplify 或 FullSimplify

In[48]:= $\left(51-14\sqrt{2}\right)^{\frac{1}{2}}-625^{\frac{1}{4}}+16^{\frac{1}{8}}$ `// FullSimplify`

Out[48]= 2

範例2 比較 $(\frac{1}{3})^{0.4}$ 、 $\frac{1}{\sqrt[6]{9}}$ 、 $\sqrt[4]{\frac{1}{27}}$ 、 $9^{-\frac{1}{4}}$

利用集合符號{ }與 //N 置換成數字比較大小

In[49]:= $\left\{\left(\frac{1}{3}\right)^{0.4}, \frac{1}{\sqrt[6]{9}}, \sqrt[4]{\frac{1}{27}}, 9^{-\frac{1}{4}}\right\}$ // N

Out[49]= {0.644394, 0.693361, 0.438691, 0.57735}

練習3 若 $a = 2^{\frac{1}{2}}, b = 3^{\frac{1}{3}}, c = 8^{\frac{1}{8}}, d = 9^{\frac{1}{9}}$

利用 {a,b,c,d}//N

In[50]:= $a = 2^{\frac{1}{2}}; b = 3^{\frac{1}{3}}; c = 8^{\frac{1}{8}}; d = 9^{\frac{1}{9}};$
{a, b, c, d} // N

Out[51]= {1.41421, 1.44225, 1.29684, 1.27652}

練習4 設 u= $\left(\sqrt{4+\sqrt{15}}\right)^{x}$, v= $\left(\sqrt{4-\sqrt{15}}\right)^{x}$ ，求 uv=?

利用 Simplify，並注意乘法要使用空白鍵！

In[52]:= $u = \left(\sqrt{4+\sqrt{15}}\right)^{x}; v = \left(\sqrt{4-\sqrt{15}}\right)^{x};$ u v // Simplify

Out[52]= 1

練習5 計算 $2^{2\log_2 3} = ?$

💡 注意 Log[n, x]的語法

$$2^{2\,\mathrm{Log}[2,3]}$$

9

練習6 計算 $\log_2 3 \log_3 7 \log_7 64$=?

```
Log[2, 3] Log[3, 7] Log[7, 64] // FullSimplify
```

6

練習7 計算 $(\log_2 5 + \log_4 0.2)\,(\log_5 2 + \log_{25} 0.5) = ?$

```
(Log[2, 5] + Log[4, 0.2]) (Log[5, 2] + Log[25, 0.5])
```

0.25

練習8 檢驗對數換底公式： $\log_a b = \frac{\log_c b}{\log_c a}$

💡 以雙等號 == 告知電腦，進行左右兩式的比較，若正確則顯示 true ，若錯誤則顯示 false。

In[53]:= $\mathrm{Log}[a, b] == \dfrac{\mathrm{Log}[c, b]}{\mathrm{Log}[c, a]}$

Out[53]= True

第六章、整數論

範例1 將 111111111（九個1）進行質因數分解

利用 FactorInteger 函數進行整數的分解：$111111111=3^2\times37^1\times333667^1$.
輸出的結果會以矩陣型態表示各因數的個數關係。

```
FactorInteger[111111111]
```

$$\begin{pmatrix} 3 & 2 \\ 37 & 1 \\ 333667 & 1 \end{pmatrix}$$

範例2 將 $2^{2^5}+1$進行質因數分解

```
FactorInteger[2^(2^5) + 1]
```

$$\begin{pmatrix} 641 & 1 \\ 6700417 & 1 \end{pmatrix}$$

範例3 第10個質數爲何？

利用 Prime 函數，可以快速找出前 10^8 個質數！

```
Prime[10]
```

29

範例4 列出前5個質數

```
Prime[{1, 2, 3, 4, 5}]
```

{2, 3, 5, 7, 11}

範例5 列出前 50 個質數

In[54]:=
```
Prime[Range[50]]
```

Out[54]= {2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113, 127, 131, 137, 139, 149, 151, 157, 163, 167, 173, 179, 181, 191, 193, 197, 199, 211, 223, 227, 229}

範例6 檢查 7919 是否爲質數。

利用 PrimeQ 函數，若是質數則輸出 true，若不是質數則輸出 false。注意Q爲大寫。

In[55]:=
```
PrimeQ[7919]
```

Out[55]= True

範例7 若 n 爲正整數，則$2^{2^n}+1$是否均爲質數？

利用 PrimeQ 函數，若是質數則輸出 true，若不是質數則輸出 false。注意Q爲大寫。

In[56]:= **Table[{$2^{2^n}+1$, PrimeQ[$2^{2^n}+1$]}, {n, 0, 5}]**

Out[56]=
$$\begin{pmatrix} 3 & \text{True} \\ 5 & \text{True} \\ 17 & \text{True} \\ 257 & \text{True} \\ 65537 & \text{True} \\ 4294967297 & \text{False} \end{pmatrix}$$

範例8 列出 120 的所有正因數

💡 利用 Divisors 函數

```
Divisors[120]
```

{1, 2, 3, 4, 5, 6, 8, 10, 12, 15, 20, 24, 30, 40, 60, 120}

{1, 2, 3, 4, 5, 6, 8, 10, 12, 15, 20, 24, 30, 40, 60, 120}

第七章、解方程式

1. 解方程式

語法　Solve[方程式 , 指定變數]

💡 利用 Solve 函數，找出「一般方程式」的解。

語法　Roots[方程式 , 指定變數]

💡 利用 Roots 函數，找出「多項方程式」的解。

語法　Reduce[方程式 , 指定變數]

💡 利用 Reduce 函數，找出「方程式所有可能的解」。

範例　解方程式 $2x^3 - 7x^2 - 7x + 30 = 0$

💡 方程式的等號必須用雙等號替代「==」, 並指定要解的未知數爲 x 。
執行完 Solve 函數後， 將以集合的型態呈現。

In[11]:= **Solve[2 x^3 - 7 x^2 - 7 x + 30 == 0, x]**

Out[11]= $\left\{\{x \to -2\}, \left\{x \to \frac{5}{2}\right\}, \{x \to 3\}\right\}$

💡 執行完 Roots 函數後， 將以熟悉的型態呈現。

In[2]:= **Roots[2 x^3 - 7 x^2 - 7 x + 30 == 0, x]**

Out[2]= $x == \frac{5}{2} \bigvee x == -2 \bigvee x == 3$

In[3]:= **Reduce[2 x^3 - 7 x^2 - 7 x + 30 == 0, x]**

Out[3]= $x == -2 \bigvee x == \frac{5}{2} \bigvee x == 3$

範例　解方程式 $x^3 - 2x^2 + 1 = 0$

In[4]:= **Solve[x^3 - 2 x^2 + 1 == 0, x]**

Out[4]= $\left\{\{x \to 1\}, \left\{x \to \frac{1}{2}\left(1 - \sqrt{5}\right)\right\}, \left\{x \to \frac{1}{2}\left(1 + \sqrt{5}\right)\right\}\right\}$

In[5]:= **Roots[x^3 - 2 x^2 + 1 == 0, x]**

Out[5]= $x == \frac{1}{2}\left(1 - \sqrt{5}\right) \bigvee x == \frac{1}{2}\left(1 + \sqrt{5}\right) \bigvee x == 1$

In[9]:= **Reduce[x^3 - 2 x^2 + 1 == 0, x]**

Out[9]= $x == 1 \bigvee x == \frac{1}{2}\left(1 - \sqrt{5}\right) \bigvee x == \frac{1}{2}\left(1 + \sqrt{5}\right)$

範例　解方程式 $x^2 - 2x + 1 = 0$

In[6]:= **Solve[x^2 - 2 x + 1 == 0, x]**

Out[6]= $\{\{x \to 1\}, \{x \to 1\}\}$

In[7]:= **Roots[x^2 - 2 x + 1 == 0, x]**

Out[7]= $x == 1 \vee x == 1$

In[8]:= **Reduce[x^2 - 2 x + 1 == 0, x]**

Out[8]= $x == 1$

範例　解方程式 $x^2 = -1$

In[10]:= **Solve[x^2 == -1, x]**

Out[10]= $\{\{x \to -i\}, \{x \to i\}\}$

In[11]:= **Roots[x^2 == -1, x]**

Out[11]= $x == i \vee x == -i$

In[12]:= **Reduce[x^2 == -1, x]**

Out[12]= $x == -i \vee x == i$

範例　解方程式 $x^3 = 1$

In[13]:= **Solve[x^3 == 1, x]**

Out[13]= $\left\{\{x \to 1\}, \left\{x \to -\sqrt[3]{-1}\right\}, \left\{x \to (-1)^{2/3}\right\}\right\}$

In[14]:= **Roots[x^3 == 1, x]**

Out[14]= $x == 1 \bigvee x == (-1)^{2/3} \bigvee x == -\sqrt[3]{-1}$

In[15]:= **Reduce[x^3 == 1, x]**

Out[15]= $x == 1 \bigvee x == -\sqrt[3]{-1} \bigvee x == (-1)^{2/3}$

💡 你將發現輸出的結果似乎不是我們想要解的樣式，可藉由面版 ComplexExpand 函數來強迫以複數 a+bi 的表示法。

In[18]:= **ComplexExpand //@ Solve[x^3 == 1, x]**

Out[18]= $\left\{\{x \to 1\}, \left\{x \to -\frac{1}{2} - \frac{i\sqrt{3}}{2}\right\}, \left\{x \to -\frac{1}{2} + \frac{i\sqrt{3}}{2}\right\}\right\}$

In[20]:= **ComplexExpand //@ Roots[x^3 == 1, x]**

Out[20]= $x == 1 \bigvee x == -\frac{1}{2} + \frac{i\sqrt{3}}{2} \bigvee x == -\frac{1}{2} - \frac{i\sqrt{3}}{2}$

In[21]:= **ComplexExpand //@ Reduce[x^3 == 1, x]**

Out[21]= $x == 1 \bigvee x == -\frac{1}{2} - \frac{i\sqrt{3}}{2} \bigvee x == -\frac{1}{2} + \frac{i\sqrt{3}}{2}$

範例　解方程式 $x^2 + x + 1 = 0$

In[22]:=
```
Solve[x^2 + x + 1 == 0, x]
```

Out[22]= $\left\{\left\{x \to -\sqrt[3]{-1}\right\}, \left\{x \to (-1)^{2/3}\right\}\right\}$

In[23]:=
```
ComplexExpand //@ Solve[x^2 + x + 1 == 0, x]
```

Out[23]= $\left\{\left\{x \to -\frac{1}{2} - \frac{i\sqrt{3}}{2}\right\}, \left\{x \to -\frac{1}{2} + \frac{i\sqrt{3}}{2}\right\}\right\}$

範例　解以 x 爲變數的二次方程式 $a\,x^2 + b\,x + c = 0$

💡 注意乘法必須以空格替代，並指定待解變數爲 x。

In[25]:=
```
Solve[a x^2 + b x + c == 0, x]
```

Out[25]= $\left\{\left\{x \to \frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a}\right\}, \left\{x \to \frac{\sqrt{b^2 - 4\,a\,c} - b}{2\,a}\right\}\right\}$

In[26]:=
```
Roots[a x^2 + b x + c == 0, x]
```

Out[26]= $x == \frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a} \bigvee x == \frac{\sqrt{b^2 - 4\,a\,c} - b}{2\,a}$

In[27]:=
```
Reduce[a x^2 + b x + c == 0, x]
```

Out[27]= $x == \frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a} \bigwedge a \neq 0 \bigvee x == \frac{\sqrt{b^2 - 4\,a\,c} - b}{2\,a} \bigwedge a \neq 0 \bigvee$
$a == 0 \wedge b == 0 \wedge c == 0 \bigvee a == 0 \bigwedge x == -\frac{c}{b} \bigwedge b \neq 0$

範例　解方程式 $x^7 - x^5 - 7\,x^3 + x^2 + 7\,x - 1 = 0$

💡 因為 5 次（含）以上的多項式都沒有公式解，因此 Solve 函數無法找到實際解的型態。

In[28]:= `Solve[x^7 - x^5 - 7 x^3 + x^2 + 7 x - 1 == 0, x]`

Out[28]= $\{\{x \to -1\}, \{x \to 1\}, \{x \to \text{Root}[\#1^5 - 7\,\#1 + 1\,\&, 1]\}, \{x \to \text{Root}[\#1^5 - 7\,\#1 + 1\,\&, 2]\}, \{x \to \text{Root}[\#1^5 - 7\,\#1 + 1\,\&, 3]\}, \{x \to \text{Root}[\#1^5 - 7\,\#1 + 1\,\&, 4]\}, \{x \to \text{Root}[\#1^5 - 7\,\#1 + 1\,\&, 5]\}\}$

💡 改用 NSolve 函數，強迫以小數近似值來表示所有的解。

In[9]:= `NSolve[x^7 - x^5 - 7 x^3 + x^2 + 7 x - 1 == 0, x]`

Out[9]= $\{\{x \to -1.66049\}, \{x \to -1.\}, \{x \to -0.0355442 - 1.62852\,i\}, \{x \to -0.0355442 + 1.62852\,i\}, \{x \to 0.142866\}, \{x \to 1.\}, \{x \to 1.58871\}\}$

In[12]:= `Solve[x^7 - x^5 - 7 x^3 + x^2 + 7 x - 1 == 0, x] // N`

Out[12]= $\{\{x \to -1.\}, \{x \to 1.\}, \{x \to -1.66049\}, \{x \to 0.142866\}, \{x \to 1.58871\}, \{x \to -0.0355442 - 1.62852\,i\}, \{x \to -0.0355442 + 1.62852\,i\}\}$

In[22]:= `Roots[x^7 - x^5 - 7 x^3 + x^2 + 7 x - 1 == 0, x] // N`

Out[22]= $x == -1.66049 \lor x == -1. \lor x == -0.0355442 - 1.62852\,i \lor x == -0.0355442 + 1.62852\,i \lor x == 0.142866 \lor x == 1. \lor x == 1.58871$

In[21]:= `Reduce[x^7 - x^5 - 7 x^3 + x^2 + 7 x - 1 == 0, x] // N`

Out[21]= $x == -1. \lor x == 1. \lor -1.66049 == x \lor 0.142866 == x \lor 1.58871 == x \lor -0.0355442 - 1.62852\,i == x \lor -0.0355442 + 1.62852\,i == x$

範例　求同時滿足兩個方程式 2x+y=1 與 x-3y=5 的解。

💡 利用 Solve 函數來解連立方成組。方程式必須當成集合來處理，另外要改成兩個變數。

In[13]:= `Solve[{2 x + y == 1, x - 3 y == 5}, {x, y}]`

Out[13]= $\left\{\left\{x \to \frac{8}{7}, y \to -\frac{9}{7}\right\}\right\}$

💡 亦可使用數學「且」的符號「∧」來連接兩個方程式。

In[12]:= `Solve[2 x + y == 1 ∧ x - 3 y == 5, {x, y}]`

Out[12]= $\left\{\left\{x \to \frac{8}{7}, y \to -\frac{9}{7}\right\}\right\}$

💡 Roots 函數只能解單一方程式，不能解聯立方程組。

```
In[14]:= Roots[{2 x + y == 1, x - 3 y == 5}, {x, y}]
```

General::ivar : $\{x, y\}$ is not a valid variable.

Out[14]= Roots[{2 x + y == 1, x − 3 y == 5}, {x, y}]

💡 Reduce 函數可以解聯立方程組。

```
In[15]:= Reduce[{2 x + y == 1, x - 3 y == 5}, {x, y}]
```

Out[15]= $x == \frac{8}{7} \bigwedge y == -\frac{9}{7}$

```
In[13]:= Reduce[2 x + y == 1 ∧ x - 3 y == 5, {x, y}]
```

Out[13]= $x == \frac{8}{7} \bigwedge y == -\frac{9}{7}$

範例　解聯立方程組 $\begin{cases} 3x - 2y + 7z = 80 \\ 5x + 3y - 4z = 2 \\ 2x + 5y + z = 42 \end{cases}$

```
In[23]:= Solve[{3 x - 2 y + 7 z == 80, 5 x + 37 - 4 z == 2, 2 x + 5 y + z == 42}, {x, y, z}]
```

Out[23]= $\left\{\left\{x \to \frac{641}{261}, y \to \frac{1319}{261}, z \to \frac{3085}{261}\right\}\right\}$

```
In[14]:= Solve[3 x - 2 y + 7 z == 80 ∧ 5 x + 37 - 4 z == 2 ∧ 2 x + 5 y + z == 42, {x, y, z}]
```

Out[14]= $\left\{\left\{x \to \frac{641}{261}, y \to \frac{1319}{261}, z \to \frac{3085}{261}\right\}\right\}$

```
In[15]:= Reduce[{3 x - 2 y + 7 z == 80, 5 x + 37 - 4 z == 2, 2 x + 5 y + z == 42}, {x, y, z}]
```

Out[15]= $x == \frac{641}{261} \bigwedge y == \frac{1319}{261} \bigwedge z == \frac{3085}{261}$

```
In[16]:= Reduce[3 x - 2 y + 7 z == 80 ∧ 5 x + 37 - 4 z == 2 ∧ 2 x + 5 y + z == 42, {x, y, z}]
```

Out[16]= $x == \frac{641}{261} \bigwedge y == \frac{1319}{261} \bigwedge z == \frac{3085}{261}$

2. 解不等式

```
In[1]:= << Algebra`InequalitySolve`
```

解不等式： $x(x^2-2)(x^2-3)>0$

```
In[6]:= InequalitySolve[x (x^2 - 2) (x^2 - 3) > 0, x]
```

Out[6]= $-\sqrt{3}<x<-\sqrt{2}\vee 0<x<\sqrt{2}\vee x>\sqrt{3}$

解不等式： $\frac{(x^2+x+1)(1-x)(x-2)^2}{(x^2-2)(x-3)}\le 0$

```
In[7]:= InequalitySolve[(x^2 + x + 1) (1 - x) (x - 2)^2 / ((x^2 - 2) (x - 3)) ≤ 0, x]
```

Out[7]= $x<-\sqrt{2}\vee 1\le x<\sqrt{2}\vee x==2\vee x>3$

解不等式： $\frac{x}{|x-1|}\ge 0$ 且 $\frac{1}{x}<x+1$

```
In[68]:= InequalitySolve[x / Abs[x - 1] ≥ 0 ∧ 1/x < x + 1, x]
```

Out[68]= $\frac{1}{2}\left(-1+\sqrt{5}\right)<x<1\bigvee x>1$

解不等式： $x^2+y^2<1$且 $x<y$

```
In[70]:= InequalitySolve[x^2 + y^2 < 1 && x < y, {x, y}]
```

Out[70]= $-1<x\le-\frac{1}{\sqrt{2}}\bigwedge -\sqrt{1-x^2}<y<\sqrt{1-x^2}\bigvee -\frac{1}{\sqrt{2}}<x<\frac{1}{\sqrt{2}}\bigwedge x<y<\sqrt{1-x^2}$

解不等式： $|x-1| \le 5$且 $e^x \le 3$

In[72]:=
```
InequalitySolve[Abs[x - 1] ≤ 5 && E^x ≤ 3, x]
```

InequalitySolve::npi :
A nonpolynomial equation or inequality encountered. The solution set may be incorrect.

Out[72]= $-4 \le x \le \log(3)$

第八章、自訂函數

語法　f[x_]:=以 x 爲變數的函數。

💡 定義 f 的函數關係。

語法　Clear[函數或變數]

💡 清除之前的函數定義或變數。

範例　若 $f(x)=2x^3+3x^2-18x+8$，則(1) f(101)-f(100)=? (2) 解 f(x)=0

In[68]:=
```
f[x_] := 2 x^3 + 3 x^2 - 18 x + 8
f[101] - f[100]
Solve[f[x] == 0, x]
Reduce[f[x] == 0, x]
```

Out[69]= 61187

Out[70]= $\{\{x \to -4\}, \{x \to \frac{1}{2}\}, \{x \to 2\}\}$

Out[71]= $x == -4 \bigvee x == \frac{1}{2} \bigvee x == 2$

範例　承上題，展開 f(x+2y)

In[73]:=
```
f[x + 2 y] // Expand
```

Out[73]= $2x^3 + 12\,y\,x^2 + 3\,x^2 + 24\,y^2\,x + 12\,y\,x - 18\,x + 16\,y^3 + 12\,y^2 - 36\,y + 8$

範例　移除 f(x) 的定義。

In[74]:=
```
Clear[f]
```

In[75]:=
```
f[1]
```

Out[75]= $f(1)$

範例　試比較 $\log_2 0.1,\ \log_2 0.5,\ \log_2 1,\ \log_2 2$ 的大小。

In[80]:=
```
f[x_] := Log[2, x]
{f[0.1], f[0.5], f[1], f[2]}
```

Out[81]= {−3.32193, −1., 0, 1}

In[82]:=
```
{0.1, 0.5, 1, 2} // f
```

Out[82]= {−3.32193, −1., 0, 1}

In[83]:=
```
Clear[f]
```

練習　試比較 $\log_{0.3} 2,\ \log_{0.8} 2,\ \log_3 2,\ \log_8 2$ 的大小。

In[86]:=
```
g[x_] := Log[x, 2]
{g[0.3], g[0.8], g[3], g[8]} // N
```

Out[87]= {−0.575717, −3.10628, 0.63093, 0.333333}

In[89]:=
```
{0.3, 0.8, 3, 8} // g // N
```
Out[89]= $\{-0.575717, -3.10628, 0.63093, 0.333333\}$

In[90]:=
```
Clear[g]
```

範例　若 $f(x)=\begin{cases} 2x+1 & , x\ge 1 \\ x^2+2 & , -2\le x<1 \\ x+8 & , x<-2 \end{cases}$，則 f(-3)=?, f(-2)=?, f(0)=?, f(2)=?

分段定義函數，其中「/;」代表條件限制。

In[4]:=
```
f[x_] := 2 x + 1 /; x ≥ 1
f[x_] := x^2 + 2 /; -2 ≤ x < 1
f[x_] := x + 8 /; x < -2
{f[-3], f[-2], f[0], f[2]}
```
Out[7]= {5, 6, 2, 5}

在 Mathematica5.1 提供了 **Piecewise**的片段函數，快速鍵為「<ESC> + <pw> + <ESC>」
利用「<Ctrl>+<ENTER>」增加下一列

In[21]:=
```
Piecewise[{{2 x + 1, x ≥ 1}, {x^2 + 2, -2 ≤ x < 1}, {x + 8, x < -2}}]
```
Out[21]=
```
Piecewise[{{1 + 2 x, x ≥ 1}, {2 + x^2, -2 ≤ x < 1}, {8 + x, x < -2}}]
```

第九章、向量

向量的加減法與係數積，若 $\vec{a}=(1,2)$, $\vec{b}=(3,4)$ ，試求 $\vec{a}+\vec{b}$, $\vec{a}-\vec{b}$, $3\vec{a}$, $\vec{a}+(3,3)$

💡 座標向量 (a,b) 必須將小刮號改成大刮號表示：{a,b}

In[16]:=
```
a = {1, 2}; b = {3, 4};
a + b
a - b
3 a
a + 3
```

Out[17]= {4, 6}

Out[18]= {−2, −2}

Out[19]= {3, 6}

Out[20]= {4, 5}

範例　若 A(0,-1), B(2,3), C(1,5), 則 ΔABC 的重心座標爲？

💡 因爲 C 的預設值爲不定積分後所產生的常數，因此要避開 C 的代號。

In[5]:=
```
{A, B, CC} = {{0, -1}, {2, 3}, {1, 5}};
```
$$\frac{A + B + CC}{3}$$

Out[6]= $\left\{1, \frac{7}{3}\right\}$

💡 亦可自行定義新的重心座標函數 OG ，以方便直接帶入使用

In[7]:= $OG[p1\_, p2\_, p3\_] := \frac{p1 + p2 + p3}{3}$

In[8]:=
```
{A, B, CC} = {{0, -1}, {2, 3}, {1, 5}};
OG[A, B, CC]
```

Out[9]= $\left\{1, \frac{7}{3}\right\}$

練習　若 A(0,-1,0), B(1,1,1), C(2,1,-2), 則 ΔABC 的重心座標爲？

In[10]:=
```
OG[{0, -1, 0}, {1, 1, 1}, {2, 1, -2}]
```

Out[10]= $\left\{1, \frac{1}{3}, -\frac{1}{3}\right\}$

範例　若 $\vec{a}=(1,2)$, $\vec{b}=(4,-2)$，求 $\vec{a}\cdot\vec{b}$, $|\vec{a}|$, $\left|2\vec{b}-3\vec{a}\right|$

內積符號為一點「.」
自行定義向量長度 norm 函數（根據 $|\vec{x}|^2=\vec{x}\cdot\vec{x}$）

In[21]:=
```
a = {1, 2}; b = {4, -2};
norm[x_] := √x.x
a.b
norm[a]
norm[2 b - 3 a]
```

Out[23]= 0

Out[24]= $\sqrt{5}$

Out[25]= $5\sqrt{5}$

練習　若 $\vec{a}=(1,2,3)$, $\vec{b}=(4,5,6)$，求 $\vec{a}\cdot\vec{b}$, $|\vec{a}|$, $\left|\vec{b}-4\vec{a}\right|$

In[26]:=
```
a = {1, 2, 3}; b = {4, 5, 6};
a.b
norm[a]
norm[b - 4 a]
```

Out[27]= 32

Out[28]= $\sqrt{14}$

Out[29]= $3\sqrt{5}$

範例　若 $\vec{a}=(1,2)$, $\vec{b}=(4,-2)$，求 $\vec{a}$ 與 $\vec{b}$ 的夾角。

利用 $\cos\theta=\frac{\vec{a}\cdot\vec{b}}{|\vec{a}||\vec{b}|}$，即 $\theta=\cos^{-1}\left(\frac{\vec{a}\cdot\vec{b}}{|\vec{a}||\vec{b}|}\right)$

In[30]:=
$$\texttt{angle[a\_, b\_] := ArcCos}\left[\frac{\texttt{a.b}}{\texttt{norm[a] norm[b]}}\right]$$

In[31]:=
```
a = {1, 2}; b = {4, -2};
angle[a, b]
```
Out[32]= $\frac{\pi}{2}$

練習　若 $\vec{a}=(1,2,3)$, $\vec{b}=(-1,-2,-3)$，求 $\vec{a}$ 與 $\vec{b}$ 的夾角。

In[33]:=
```
angle[{1, 2, 3}, {-1, -2, -3}]
```
Out[33]= π

範例　若 $\vec{a}=(1,2)$, $\vec{b}=(3,4)$，求 $\vec{a}$ 在 $\vec{b}$ 上的正射影。

In[34]:=
```
a = {1, 2}; b = {3, 4};
```
$\frac{\texttt{a.b}}{\texttt{b.b}}\ \texttt{b}$

Out[35]= $\left\{\frac{33}{25}, \frac{44}{25}\right\}$

範例　若 $\vec{a}=(1,2,3)$, $\vec{b}=(4,5,6)$，計算 $\vec{a}\times\vec{b}$

利用面版「×」符號，或者直接輸入快速鍵「<ESC>cross<ESC>」

In[38]:=
```
{1, 2, 3} × {4, 5, 6}
```
Out[38]= $\{-3, 6, -3\}$

範例　一直線 L 與向量 (1,-3) 垂直，且通過點(1,2)，試求此直線方程式。

In[42]:=
```
{1, -3}.({x, y} - {1, 2}) == 0
```
Out[42]= $x-3(y-2)-1==0$

範例　解聯立方程組$\begin{cases} 3x-2y+7z=80 \\ 5x+3y-4z=2 \\ 2x+5y+z=42 \end{cases}$

```
In[83]:= L1 = {3, -2, 7}.{x, y, z} == 80;
         L2 = {5, 3, -4}.{x, y, z} == 2;
         L3 = {2, 5, 1}.{x, y, z} == 42;
         Solve[{L1, L2, L3}, {x, y, z}]
         Reduce[{L1, L2, L3}, {x, y, z}]
```

Out[86]= $\{\{x \to 6,\ y \to 4,\ z \to 10\}\}$

Out[87]= $x == 6 \land y == 4 \land z == 10$

```
In[11]:= {2 x + 3 y - 2 z == 5}[[1]][[1]]
```

Out[11]= $2x+3y-2z$

平面方程式：$2x+3y-2z=5$ 的法向量為何？

```
In[8]:= Coefficient[{2 x + 3 y - 2 z == 5}[[1]][[1]], {x, y, z}]
```

Out[8]= $\{2, 3, -2\}$

回家作業　寫出求內心函數 OI、外心函數 OT、垂心函數OH，給平面（或空間）中的非共線三點，即可求出內心座標。

例如 OI[p1,p2,p3] 為p1,p2,p3的內心座標、OT[p1,p2,p3] 為p1,p2,p3的外心座標、OH[p1,p2,p3] 為p1,p2,p3的垂心座標。

第十章、矩陣

矩陣輸入　利用面版 $\begin{pmatrix} \square & \square \\ \square & \square \end{pmatrix}$ 來輸入矩陣。

增加一行 <ctrl>+<,>

增加一列 <ctrl>+<Enter>

MatrixPower[A, n]　計算 A^n

A. B　計算矩陣乘法 AB

Transpose[A]　計算轉置矩陣 A^T

Det[A]　計算行列式 |A|

Inverse[A]　計算反矩陣 A^{-1}

Tr[A]　計算方陣的主對角線和

範例　若 A= $\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 3 & -1 & 2 \end{pmatrix}$, B= $\begin{pmatrix} -1 & 2 & -2 \\ 2 & -2 & -3 \\ 1 & 4 & 1 \end{pmatrix}$，計算 A+B, A^2 , AB, |A|, A^{-1}, A的主對角線和

矩陣相乘必須用「.」，絕不可以使用「空白鍵」！！
反矩陣 必須使用「Inverse」，絕不可以使用「A^{-1}」！！
行列式必須使用「Det」，絕不可以使用「|A|」！！
n次方必須使用「Power[A, n]」，絕不可以使用「 A^n 」

In[77]:= $A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 3 & -1 & 2 \end{pmatrix}$; $B = \begin{pmatrix} -1 & 2 & -2 \\ 2 & -2 & -3 \\ 1 & 4 & 1 \end{pmatrix}$;

```
A + B
Power[A, 2]
A.B
Det[A]
Inverse[A]
Tr[A]
```

Out[78]= $\begin{pmatrix} 0 & 4 & 1 \\ 6 & 3 & 3 \\ 4 & 3 & 3 \end{pmatrix}$

Out[79]= $\begin{pmatrix} 1 & 4 & 9 \\ 16 & 25 & 36 \\ 9 & 1 & 4 \end{pmatrix}$

Out[80]= $\begin{pmatrix} 6 & 10 & -5 \\ 12 & 22 & -17 \\ -3 & 16 & -1 \end{pmatrix}$

Out[81]= -21

Out[82]= $\begin{pmatrix} -\frac{16}{21} & \frac{1}{3} & \frac{1}{7} \\ -\frac{10}{21} & \frac{1}{3} & -\frac{2}{7} \\ \frac{19}{21} & -\frac{1}{3} & \frac{1}{7} \end{pmatrix}$

Out[83]= 8

範例　試化簡 $\begin{vmatrix} 1 & a & a^2 \\ 1 & b & b^2 \\ 1 & c & c^2 \end{vmatrix}$

In[84]:= $\mathrm{Det}\left[\begin{pmatrix} 1 & a & a^2 \\ 1 & b & b^2 \\ 1 & c & c^2 \end{pmatrix}\right]$ // Factor

Out[84]= $-(a-b)(a-c)(b-c)$

範例　解聯立方程組 $\begin{cases} x+2y+3z=2 \\ 4x+5y+6z=1 \\ 3x-y+2z=0 \end{cases}$

In[95]:= `Solve[{x + 2 y + 3 z == 2, 4 x + 5 y + 6 z == 1, 3 x - y + 2 z == 0}, {x, y, z}]`

Out[95]= $\left\{\left\{x \to -\frac{25}{21}, y \to -\frac{13}{21}, z \to \frac{31}{21}\right\}\right\}$

In[96]:= `Solve[` $\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 3 & -1 & 2 \end{pmatrix} . \begin{pmatrix} x \\ y \\ z \end{pmatrix} == \begin{pmatrix} 2 \\ 1 \\ 0 \end{pmatrix}$`, {x, y, z}]`

Out[96]= $\left\{\left\{x \to -\frac{25}{21}, y \to -\frac{13}{21}, z \to \frac{31}{21}\right\}\right\}$

In[98]:= `LinearSolve[` $\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 3 & -1 & 2 \end{pmatrix}, \begin{pmatrix} 2 \\ 1 \\ 0 \end{pmatrix}$`]`

Out[98]= $\begin{pmatrix} -\frac{25}{21} \\ -\frac{13}{21} \\ \frac{31}{21} \end{pmatrix}$

第十一章、線性規劃

語法	Minimize[{目標, 限制條件1, 限制條件2, ..., 限制條件n} , {變數}]	找出最小值
	Maximize[{目標, 限制條件1, 限制條件2, ..., 限制條件n} , {變數}]	找出最大值
	NMinimize[{目標, 限制條件1, 限制條件2, ..., 限制條件n} , {變數}]	找出最小值數值解
	NMaximize[{目標, 限制條件1, 限制條件2, ..., 限制條件n} , {變數}]	找出最大值數值解

這是 Mathematica5.0 版以後才提供的功能，可以求非線性的限制條件與非線性的目標！！

範例　限制條件： $x+y\le 10$ 、$x-y\ge 2$ 、$x\ge 1$ 　目標條件：求 $2x-3y$ 的最小值。

💡 利用 ConstrainedMin 函數解線性規劃問題

```
In[10]:= Minimize[{2 x - 3 y, x + y ≤ 10, x - y ≥ 2, x ≥ 1}, {x, y}]
Out[10]= {0, {x → 6, y → 4}}
```

練習　限制條件：x-3 y≤7,2 x+3 z≥5,x+y+z≤10　目標條件：求x+3 y+7 z的最小值

```
In[18]:= Minimize[{x + 3 y + 7 z, x - 3 y ≤ 7, 2 x + 3 z ≥ 5, x + y + z ≤ 10}, {x, y, z}]
Out[18]= {-38, {x → 16, y → 3, z → -9}}
```

練習　限制條件： $x^2 + y^2 \le 1$　目標條件：求 $\mathrm{Cos}(x + 2\, y)$ 的最小值

💡 由於 $x^2 + y^2 \le 1$ 並非線性不等式，所以無法使用「Minimize」解之

```
In[23]:= Minimize[{Cos[x + 2 y], x^2 + y^2 ≤ 1}, {x, y}]
Out[23]= Minimize[{Cos[x + 2 y], x^2 + y^2 ≤ 1}, {x, y}]
```

💡 改由「**NMinimize**」解之即可！

```
In[24]:= NMinimize[{Cos[x + 2 y], x^2 + y^2 ≤ 1}, {x, y}]
Out[24]= {-0.617273, {x → 0.447214, y → 0.894427}}
```

範例　有 A, B兩種食品，A食品含有蛋白質10%、脂肪20%、碳水化合物40%，B食品含有蛋白質15%、脂肪8%、碳水化合物20%。A食品的價格爲每100公克44元，B食品的價格爲每100公克55元。若一個人每天營養中最少需要蛋白質45公克、脂肪50.4公克，碳水化合物則沒有限制。試問某人每天必須食用A食品多少公克？B食品多少公克？才能有足夠的營養，且最省錢？

假設某人每天必須食用A食品 x 公克、B食品 y 公克
限制條件：10% x + 15% y ≥ 45，20% x +8% y ≥50.4，x≥0，y≥0
目標： $44\frac{x}{100}+55\frac{y}{100}$ 爲最小値

In[34]:= $\text{Minimize}\left[\left\{44\frac{x}{100}+55\frac{y}{100},\ \frac{10}{100}x+\frac{15}{100}y\geq 45,\ \frac{20}{100}x+\frac{8}{100}y\geq 50.4,\ x\geq 0,\ y\geq 0\right\},\ \{x,\ y\}\right]$

Out[34]= {178.2, {x → 180., y → 180.}}

以下指令將於「第十三章、繪圖」中介紹
匯入不等式套件指令 <<Graphics`InequalityGraphics`

In[28]:= << Graphics`InequalityGraphics`

In[40]:= $\text{InequalityPlot}\left[\left\{\frac{10}{100}x+\frac{15}{100}y\geq 45,\ \frac{20}{100}x+\frac{8}{100}y\geq 50.4,\ x\geq 0,\ y\geq 0\right\},\ \{x,\ 0,\ 1000\},\ \{y,\ 0,\ 1000\},\ \text{AxesOrigin}\rightarrow\{0,\ 0\}\right]$



Out[40]= - Graphics -

第十二章、簡單的微分

語法　以下將 f(x) 裡所有非 x 的符號均視爲常數來計算
D[f, x]　將 f 對 x 變數進行微分
D[f, {x,n}]　將 f 對 x 變數進行 n 次微分
f'[x]　將 f 對 x 變數進行微分
f''[x]　將 f 對 x 變數進行二次微分微分
f'''[x]　將 f 對 x 變數進行三次微分微分
$\partial_x f[x]$　將 f 對 x 變數進行微分
$\partial_{\{x,n\}} f[x]$　將 f 對 x 變數進行 n 次微分

範例　計算 $\frac{d}{dx}(2x^3 - x^2 + x - 2)$

In[62]:= `D[2 x^3 - x^2 + x - 2, x]`

Out[62]= $6x^2 - 2x + 1$

In[63]:= `∂x (2 x^3 - x^2 + x - 2)`

Out[63]= $6x^2 - 2x + 1$

練習　計算 sin'(x)

In[65]:= `Sin'[x]`

Out[65]= $\cos(x)$

In[64]:= `∂x Sin[x]`

Out[64]= $\cos(x)$

In[66]:= `D[Sin[x]]`

Out[66]= $\sin(x)$

範例　計算 $\frac{d}{dx}(f(x)+g(x))$ 、$\frac{d}{dx}(f(x)-g(x))$、$\frac{d}{dx}(f(x)\,g(x))$、$\frac{d}{dx}\frac{f(x)}{g(x)}$ 、$\frac{d}{dx}f(g(x))$

In[71]:= `D[f[x] + g[x], x]`

Out[71]= $f'(x) + g'(x)$

In[72]:= `D[f[x] - g[x], x]`

Out[72]= $f'(x) - g'(x)$

In[73]:= `D[f[x] g[x], x]`

Out[73]= $g(x)\,f'(x) + f(x)\,g'(x)$

In[74]:= $D\left[\frac{f[x]}{g[x]}, x\right]$

Out[74]= $\frac{f'(x)}{g(x)} - \frac{f(x)\, g'(x)}{g(x)^2}$

In[83]:= `D[f[g[x]], x]`

Out[83]= $f'(g(x))\, g'(x)$

練習　若 f(x)= $x^2 \sin(x)$ ，計算 f'(x)

In[67]:= `f[x_] := x^2 Sin[x]`

In[68]:= `f'[x]`

Out[68]= $\cos(x)\, x^2 + 2 \sin(x)\, x$

語法　將 f(x) 裡所有符號均視爲x的函數來計算全微分 $\frac{d}{dx} f(x)$
Dt[f, x]　將 f 對 x 變數進行全微分
Dt[f, {x,n}]　將 f 對 x 變數進行 n 次全微分

範例　計算全微分 $\frac{d}{dx}(x^2 + y)$

In[76]:= `Dt[x^2 + y, x]`

Out[76]= $2\, x + \frac{dy}{dx}$

第十三章、繪圖

1. 2D基本繪圖

語法	Plot[f(x) , {x,min,max}]	繪製 y=f(x) 的圖形，並限制 x 的範圍
	Plot[{f(x),g(x)},{x,min,max}]	繪製 y=f(x) 與 y=g(x) 的圖形
	Plot[{f(x),g(x),h(x)},{x,min,max}]	繪製 y=f(x) 、 y=g(x) 與 y=h(x) 的圖形

範例　繪製 $y = x^2,\ -1 \le x \le 1$ 的圖形。

In[88]:= **Plot[x^2, {x, -1, 1}]**



Out[88]= - Graphics -

範例　繪製 $y = 2\,x^3 - x^2 + 1,\ -1 \le x \le 1$ 的圖形。

In[93]:= **Plot[2 x^3 - x^2 + 1, {x, -1, 1}]**



Out[93]= - Graphics -

練習　繪製 $y = \sin x,\ 0 \le x \le 2\pi$ 的圖形。

In[89]:= **Plot[Sin[x], {x, 0, 2 π}]**



Out[89]= - Graphics -

練習　繪製 $y = \cos x,\ -2\pi \le x \le 2\pi$ 的圖形。

In[90]:=
```
Plot[Cos[x], {x, -2 π, 2 π}]
```



Out[90]= - Graphics -

練習　繪製 $y = \log_2 x,\ 0 \le x \le 5$ 的圖形。

In[91]:=
```
Plot[Log[2, x], {x, 0, 5}]
```



Out[91]= - Graphics -

範例　同時繪製 $y = \sin x$與 $y = \cos x$, $0 \le x \le 2\pi$的圖形

In[94]:=
```
Plot[{Sin[x], Cos[x]}, {x, 0, 2π}]
```



Out[94]= - Graphics -

2.繪圖樣式

In[109]:=
```
Plot[Sin[x], {x, -6π, 6π}, Ticks → {{π, 0, 2π, 3π}, {1, -1}}]
```



Out[109]= - Graphics -

In[113]:=
```
Plot[Sin[x], {x, -6π, 6π},
  Ticks → {{{π, "kk"}, {2π, "2π"}}, Automatic}]
```



Out[113]= - Graphics -

3. 繪圖顏色

語法　Plot[f(x), {x, xmin, xmax}, PlotStyle→RGBColor[a,b,c]]

繪製f(x)的圖形，並指定 x 從 xmin 到xmax，且圖形顏色爲RGBColor[a,b,c]

R代表紅色系,　　G代表綠色系,　　B代表藍色系　　0≤a, b, c≤1

RGBColor[1,0,0] 代表紅色

RGBColor[0,0,1] 代表藍色

RGBColor[0,1,0] 代表綠色

RGBColor[1,1,1] 代表白色

RGBColor[0,0,0] 代表黑色

RGBColor[1,1,0] 代表黃色

RGBColor[0,1,1] 代表淡藍色

RGBColor[1,0,1] 代表紫色

範例　以紅色繪製 $y = \sin x,\ 0 \le x \le 2\pi$ 的圖形。

In[95]:=
```
Plot[Sin[x], {x, 0, 2π}, PlotStyle→RGBColor[1, 0, 0]]
```



Out[95]= - Graphics -

範例　以紅色繪製 $y=\sin x$，並以藍色繪製 $y=\cos x,\ 0\le x\le 2\pi$的圖形

```
In[96]:= Plot[{Sin[x], Cos[x]}, {x, 0, 2 π},
           PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```




```
Out[96]= - Graphics -
```

範例　若 $f(x)=x^3$，試比較 y=f(x) 、 y=f(x+1) 與 y=f(x)+1 圖形之間的關係。

```
In[97]:= f[x_] := x^3
         Plot[{f[x], f[x + 1], f[x] + 1}, {x, -3, 3}, PlotStyle →
           {RGBColor[0, 0, 0], RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```




```
Out[98]= - Graphics -
```

範例　$y=\log_2 x$ 與 $y=x-1$ 有幾個交點？

因爲 $\log_2 x$ 只有在 x>0 時才有定義，因此會出現警告訊息，提醒你 $y=\log_2 x$圖形可能要注意！

In[104]:=
```
Plot[{Log[2, x], x - 1}, {x, -10, 10},
  PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```

Plot::plnr : $\log_2(x)$ is not a machine−size real number at $x=-10.$.

Plot::plnr : $\log_2(x)$ is not a machine−size real number at $x=-9.18866$.

Plot::plnr : $\log_2(x)$ is not a machine−size real number at $x=-8.30382$.

General::stop : Further output of Plot::plnr will be suppressed during this calculation.



Out[104]= - Graphics -

💡 做一點修正，改繪製 $0 \le x \le 3$ 之間的圖形。

In[105]:=
```
Plot[{Log[2, x], x - 1}, {x, 0, 3},
 PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```



Out[105]= - Graphics -

💡 從圖形上可以發現「好像」在 $-1 \le y \le 2$ 時有出現交點，因此利用 PlotRange來修正 y 的範圍。

In[106]:=
```
Plot[{Log[2, x], x - 1}, {x, 0, 3},
 PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]},
 PlotRange → {-1, 2}]
```



Out[106]= - Graphics -

練習　圖形 $y = \sin x$ 與 $y = \frac{1}{10}x$ 有幾個交點？

In[109]:=
```
Plot[{Sin[x], 1/10 x}, {x, -4 π, 4 π},
  PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```



Out[109]= - Graphics -

練習　圖形 $y = \sin x$ 與 $y = x^2$有幾個交點？

In[111]:=
```
Plot[{Sin[x], x^2}, {x, -π, π},
  PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```



Out[111]= - Graphics -

範例　繪製 f(x)=|x-3|+|x+1| 的圖形

In[17]:=
```
f[x_] := Abs[x - 3] + Abs[x + 1]
Plot[f[x], {x, -10, 10}, PlotStyle → RGBColor[1, 0, 0]]
```



Out[18]= - Graphics -

範例　繪製 $f(x)=\begin{cases} 2x+1 & , x \ge 1 \\ x^2+2 & , -2 \le x < 1 \\ x+8 & , x < -2 \end{cases}$ 的圖形

在 Mathematica5.0 提供了 **Piecewise**的片段函數，快速鍵爲「<ESC> + <pw> + <ESC>」
利用「<Ctrl>+<ENTER>」增加下一列

In[9]:=
```
f[x_] := Piecewise[{{2 x + 1, x ≥ 1}, {x^2 + 2, -2 ≤ x < 1}, {x + 8, x < -2}}]
```

In[10]:=
```
Plot[f[x], {x, -5, 5}]
```



Out[10]= - Graphics -

◎ 回家作業◎　試討論 $y = \log_n x$ 與 $y = x^n$ 的圖形交點個數。

4、2D點圖形 ListPlot

範例　畫出 (1,1), (2,2), (3,4), (4,8), (5,16) 的點圖

In[17]:=
```
ListPlot[{{1, 1}, {2, 2}, {3, 4}, {4, 8}, {5, 16}}]
```



Out[17]= - Graphics -

💡 利用 PlotStyle→PointSize[i] 來改變點的大小

In[18]:=
```
ListPlot[{{1, 1}, {2, 2}, {3, 4}, {4, 8}, {5, 16}},
 PlotStyle → PointSize[0.02]]
```



Out[18]= - Graphics -

利用 PlotJoined→True 將點依序以線段連起來

In[21]:= `ListPlot[{{1, 1}, {2, 2}, {3, 4}, {4, 8}, {5, 16}}, PlotJoined → True]`



Out[21]= - Graphics -

範例　畫出前 100 個質數的分佈圖。

In[33]:=
```
ListPlot[Table[{i, Prime[i]}, {i, 1, 100}],
  PlotStyle → {PointSize[0.01], RGBColor[1, 0, 0]},
  AxesOrigin → {0, 0}]
```



Out[33]= - Graphics -

範例　畫出 y=sin(x), x從0到10且間距爲 0.5 的點圖。

```
In[42]:= ListPlot[Table[{i, Sin[i]}, {i, 0, 10, 0.5}],
          PlotStyle → PointSize[0.01]]
```



Out[42]= - Graphics -

```
In[44]:= ListPlot[Table[{i, Sin[i]}, {i, 0, 10, 0.5}], PlotJoined → True,
          PlotStyle → {RGBColor[1, 0, 0], Thickness[0.01]}]
```



Out[44]= - Graphics -

5、圖形合併

範例　繪製 $y=\sin x,\ y=\cos x,\ y=\tan x,\ y=x^2,\ y=x^3+x+1,\ y=\lvert x+1\rvert-\lvert x-3\rvert$圖形

In[54]:=
```
Plot[{Sin[x], Cos[x], Tan[x], x^2, x^3 + x + 1, Abs[x + 1] - Abs[x - 3]},
  {x, -10, 10}, PlotStyle →
   {RGBColor[1, 0, 0], RGBColor[0, 1, 0], RGBColor[0, 0, 1],
    RGBColor[1, 1, 0], RGBColor[1, 0, 1], RGBColor[0, 1, 1]}];
```



In[95]:=
```
g1 = Plot[Sin[x], {x, -10, 10},
    DisplayFunction → Identity, PlotStyle → RGBColor[1, 0, 0]];
g2 = Plot[Cos[x], {x, -10, 10}, DisplayFunction → Identity,
    PlotStyle → RGBColor[0, 1, 0]];
g3 = Plot[Tan[x], {x, -10, 10}, DisplayFunction → Identity,
    PlotStyle → RGBColor[0, 0, 1]];
g4 = Plot[x^2, {x, -10, 10}, DisplayFunction → Identity,
    PlotStyle → RGBColor[1, 1, 0]];
g5 = Plot[x^3 + x + 1, {x, -10, 10}, DisplayFunction → Identity,
    PlotStyle → RGBColor[1, 0, 1]];
g6 = Plot[Abs[x + 1] - Abs[x - 3], {x, -10, 10},
    DisplayFunction → Identity, PlotStyle → RGBColor[0, 1, 1]];
```

In[101]:=
```
Show[g1, Frame → True]
```

Out[101]= - Graphics -

In[110]:=
```
Show[g1, DisplayFunction → $DisplayFunction]
```



Out[110]= - Graphics -

In[112]:=
```
Show[GraphicsArray[{g1, g2}], DisplayFunction → $DisplayFunction]
```



Out[112]= - GraphicsArray -

In[114]:=
```
Show[GraphicsArray[{{g1, g2}, {g3, g4}, {g5, g6}}],
  DisplayFunction → $DisplayFunction]
```



Out[114]= - GraphicsArray -

In[115]:=
```
Show[GraphicsArray[{{g1, g2, g3}, {g4, g5, g6}}],
  DisplayFunction → $DisplayFunction]
```



Out[115]= - GraphicsArray -

6、2D參數式繪圖

語法　ParametricPlot[{f(t) , g(t)} , {t,a,b}] 繪製參數 x=f(t), y=g(t), $a \le t \le b$

範例　繪製 $x = 2 + t,\ y = -3 + 2t$ 的圖形

In[116]:= **ParametricPlot[{2 + t, -3 + 2 t}, {t, -10, 10}]**



Out[116]= - Graphics -

範例　繪製 $x = \sin t$, $y = \cos t$ 的圖形

In[112]:= **ParametricPlot[{Sin[t], Cos[t]}, {t, 0, 2 π}]**



Out[112]= - Graphics -

💡 發現上圖的比例不對，因此需要藉由 AspectRatio 來進行比例修正。

In[113]:= **ParametricPlot[{Sin[t], Cos[t]},**
{t, 0, 2 π}, AspectRatio → Automatic]



Out[113]= - Graphics -

In[123]:=
```
ParametricPlot[{Sin[t], Cos[t]}, {t, 0, 2 π},
  AspectRatio → Automatic, PlotStyle → RGBColor[1, 0, 0]]
```



Out[123]= - Graphics -

練習　繪製 $x = 3\sin t$, $y = 4\cos t$ 的圖形

原來此圖形爲上下狀，且長軸半徑爲4，短軸半徑爲3的橢圓

In[124]:=
```
ParametricPlot[{3 Sin[t], 4 Cos[t]}, {t, 0, 2 π},
  AspectRatio → Automatic, PlotStyle → RGBColor[1, 0, 0]]
```



Out[124]= - Graphics -

7、2D隱函數繪圖

In[1]:=
```
<< Graphics`ImplicitPlot`
```

必須關閉 Mathematica 軟體，並重新開啓，載入上述隱函數套件指令，才能順利進行以下的程式。

語法　ImplicitPlot[方程式, 變數範圍]

範例　繪製 $x + y = 1,\ -5 \le x \le 5$ 的圖形。

In[2]:= **ImplicitPlot[x + y == 1, {x, -5, 5}]**



Out[2]= - Graphics -

In[122]:= **ImplicitPlot[x + y == 1, {x, -5, 5}, PlotStyle → RGBColor[1, 0, 0]]**



Out[122]= - Graphics -

範例　繪製 $\frac{x^2}{25}-\frac{y^2}{16}=1,\ -20\le x\le 20$ 的圖形。

In[135]:=
```
ImplicitPlot[x^2/25 - y^2/16 == 1, {x, -20, 20},
  PlotStyle → RGBColor[1, 0, 0], AspectRatio → Automatic]
```



Out[135]= - Graphics -

範例　觀察 $\frac{x^2}{25}-\frac{y^2}{16}=1,\ \frac{x^2}{36}+\frac{y^2}{5}=1$ 兩者圖形的相交狀況。

In[134]:=
```
ImplicitPlot[{x^2/25 - y^2/16 == 1, x^2/36 + y^2/5 == 1}, {x, -10, 10},
  PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]},
  AspectRatio → Automatic]
```

$$\text{ImplicitPlot}\left[\left\{\frac{x^2}{25}-\frac{y^2}{16}==1,\ \frac{x^2}{36}+\frac{y^2}{5}==1\right\},\ \{x,-10,10\}\right]$$



Out[134]= - Graphics -

練習　觀察 $x+2y^2=3$ 與 $x+y=1$ 的圖形交點個數。

In[138]:=
```
ImplicitPlot[{x + 2 y^2 == 3, x + y == 1}, {x, -4, 4},
 PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]},
 AspectRatio → Automatic]
```



Out[138]= - Graphics -

練習　觀察 $xy = 1$ 與 $x - y = 1$ 的圖形交點個數。

發現圖形似乎不平滑，而且連續性也怪怪的。

```
In[3]:= ImplicitPlot[{x y == 1, x - y == 0}, {x, -5, 5},
         PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```




```
Out[3]= - Graphics -
```

經過 PlotPoints → n 來修正繪製的精細度。（n 的預設值爲15）
另外使用 PlotRange 來限制值域，讓圖形看起來更漂亮！

Plot3D

In[10]:=
```
ImplicitPlot[{x y == 1, x - y == 0}, {x, -5, 5},
 PlotStyle → {RGBColor[0, 0, 1], RGBColor[1, 0, 0]},
 AspectRatio → Automatic, PlotPoints → 100, PlotRange -> {-5, 5}]
```



Out[10]= - Graphics -

8、3D繪圖（z=f(x, y)）

語法	Plot3D[f(x,y), {x,xmin,xmax} , {y,ymin,ymax}]

範例	繪製 x+y+z=0 的圖形。

In[139]:= `Plot3D[-x - y, {x, -1, 1}, {y, -1, 1}]`



Out[139]= - SurfaceGraphics -

範例　繪製 z=sin(xy) 的圖形

In[140]:= `Plot3D[Sin[x y], {x, 0, 4}, {y, 0, 4}]`



Out[140]= - SurfaceGraphics -

💡 此時會看到粗糙的繪圖，因此可以使用 PlotPoints → n 來修正繪製的精細度。（n的預設值為15）

In[141]:= `Plot3D[Sin[x y], {x, 0, 4}, {y, 0, 4}, PlotPoints → 60]`



Out[141]= - SurfaceGraphics -

💡 若想要將格線除去，則可以使用 Mesh→False 來去除格線。

In[142]:= `Plot3D[Sin[x y], {x, 0, 4}, {y, 0, 4}, PlotPoints → 60, Mesh → False]`



Out[142]= - SurfaceGraphics -

```
In[1]:= Plot3D[Sin[x y], {x, 0, 4}, {y, 0, 4},
         PlotPoints → 60, Mesh → False, Boxed → False, Axes → False]
```



```
Out[1]= - SurfaceGraphics -
```

練習　繪製 $z = \sin\left(\sqrt{x^2 + y^2}\right),\ -2\pi \le x \le 2\pi,\ -2\pi \le y \le 2\pi$

```
In[2]:= Plot3D[Sin[√(x² + y²)], {x, -2 π, 2 π}, {y, -2 π, 2 π},
         PlotPoints → 60, Mesh → False, Boxed → False, Axes → False]
```



```
Out[2]= - SurfaceGraphics -
```

範例　繪製並觀察三平面的關係：
$3x-2y+7z-80=0;\ 5x+3y-4z-2=0;\ 2x+5y+z-42=0$

繪製三平面的

In[11]:=
```
z1 = Plot3D[{(-3 x + 2 y + 80)/7, RGBColor[1, 0, 0]},
    {x, 5, 7}, {y, 3, 5}, DisplayFunction → Identity];
z2 = Plot3D[{(5 x + 3 y - 2)/4, RGBColor[0, 0, 1]}, {x, 5, 7},
    {y, 3, 5}, DisplayFunction → Identity];
z3 = Plot3D[{-2 x - 5 y + 42, RGBColor[0, 1, 0]},
    {x, 5, 7}, {y, 3, 5}, DisplayFunction → Identity];
Show[z1, z2, z3, DisplayFunction → $DisplayFunction]
```



Out[14]= - Graphics3D -

9、3D參數式繪圖

範例　繪製 $x=t$，$y=\cos t$，$z=\sin t$，$-2\pi\le t\le 2\pi$的圖形

In[21]:= **ParametricPlot3D[{t, Cos[t], Sin[t]}, {t, -2 π, 2 π}];**



練習　繪製 $x = \cos 5t$, $y = \sin 3t$, $z = \sin t$, $0 \le t \le 2\pi$的圖形

In[23]:= **ParametricPlot3D[{Cos[5 t], Sin[3 t], Sin[t]}, {t, 0, 2 π}];**



10、不等式繪圖

語法　InequalityPlot[不等式, 變數1範圍, 變數2範圍]

匯入不等式套件指令 <<Graphics`InequalityGraphics`

```
In[27]:= << Graphics`InequalityGraphics`
```

範例　試畫出 $1 \le (x+2y)^2 + 4y^2 \le 4$的圖形

```
In[2]:= InequalityPlot[1 ≤ (x + 2 y)^2 + 4 y^2 ≤ 4, {x, -3, 3}, {y, -5, 5}];
```



因爲此圖形有上下界，與左右界，因此對於 x, y 可以不限制範圍。

```
In[21]:= InequalityPlot[1 ≤ (x + 2 y)^2 + 4 y^2 ≤ 4, {x}, {y}];
```



範例　試作滿足 $1 < x^2 - y^2 < 4,\ \mathrm{xy} < 1,\ x > 0,\ y > 0.$ 等條件的圖形區域。

利用且的符號「∧」來連接條件

In[17]:= `InequalityPlot[1 < x^2 - y^2 < 4 ∧ x y < 1 ∧ x > 0 ∧ y > 0, {x, -2, 3}, {y, -2, 2}, AspectRatio → Automatic]`



Out[17]= - Graphics -

因爲此圖形有上下界，與左右界，因此對於 x, y 可以不限制範圍。

In[20]:= `InequalityPlot[1 < x^2 - y^2 < 4 ∧ x y < 1 ∧ x > 0 ∧ y > 0, {x}, {y}, AspectRatio → Automatic]`



Out[20]= - Graphics -

範例　試作出 $||x|-|y|| \le 1$ 的區域圖形。

由於 x, y 是無限延伸的圖形，無法抓住全貌，因此限制只畫出區域圖形

In[18]:= `InequalityPlot[Abs[Abs[x] - Abs[y]] ≤ 1, {x, -2, 2}, {y, -2, 2}]`



Out[18]= - Graphics -

範例　試作出 $z \in \mathbb{C}$, $|z| \le 1$的複數平面圖形

In[55]:= `ComplexInequalityPlot[ Abs[z] ≤ 1, {z} ];`



範例　試作出 $z \in \mathbb{C}$，$|z^3 + 3z^2 - (1+2i)|^2 \le |z+i|^2$. 的複數平面圖形

In[56]:= `ComplexInequalityPlot[ Abs[ z^3 + 3 z^2 - (1 + 2 i) ]^2 ≤ Abs[z + i]^2, {z} ];`



專題研究

norm 向量長度

In[1]:=
```
norm[v_] := √v.v
norm[A_, B_] := norm[A - B]
```

area 三點求面積

In[3]:=
```
area[p1_, p2_, p3_] := Abs[Det[{p2 - p1, p3 - p1}]] / 2
```

OI 三點求內心

```
In[4]:= OI[p1_, p2_, p3_] := Module[{a, b, c}, a = norm[p2, p3];
          b = norm[p1, p3]; c = norm[p1, p2]; (a p1 + b p2 + c p3)/(a + b + c)];
        OI[{a1_, b1_, c1_}, {a2_, b2_, c2_}, {a3_, b3_, c3_}] :=
         Module[{aa1, bb1, cc1, aa2, bb2, cc2, aa3, bb3, cc3,
           n12, n23, n31, , L1, L2, L3, M1, M2, a, b, c, d, e, f, r},
          L1 = {aa1, bb1, cc1} = norm[{a2, b2}] norm[{a3, b3}] {a1, b1, c1};
          L2 = {aa2, bb2, cc2} = norm[{a1, b1}] norm[{a3, b3}] {a2, b2, c2};
          L3 = {aa3, bb3, cc3} = norm[{a1, b1}] norm[{a2, b2}] {a3, b3, c3};
          n12 = {aa1, bb1}.{aa2, bb2};
          n23 = {aa2, bb2}.{aa3, bb3}; n31 = {aa3, bb3}.{aa1, bb1};
          If[Abs[n12] ≥ Abs[n23] && Abs[n31] ≥ Abs[n23], If[n12 > 0,
            M1 = L1 + L2, M1 = L1 - L2]; If[n31 > 0, M2 = L3 + L1, M2 = L3 - L1];
           r = Flatten[Solve[{M1.{x, y, 1} == 0, M2.{x, y, 1} == 0}, {x, y}]]];
          If[Abs[n12] ≥ Abs[n31] && Abs[n23] ≥ Abs[n31], If[n12 > 0,
            M1 = L1 + L2, M1 = L1 - L2]; If[n23 > 0, M2 = L2 + L3, M2 = L2 - L3];
           r = Flatten[Solve[{M1.{x, y, 1} == 0, M2.{x, y, 1} == 0}, {x, y}]]];
          If[Abs[n23] ≥ Abs[n12] && Abs[n31] ≥ Abs[n12], If[n31 > 0,
            M1 = L3 + L1, M1 = L3 - L1]; If[n23 > 0, M2 = L2 + L3, M2 = L2 - L3];
           r = Flatten[Solve[{M1.{x, y, 1} == 0, M2.{x, y, 1} == 0}, {x, y}]]]]
```

r 三點求內切圓半徑

```
In[6]:= r[p1_, p2_, p3_] := (2 × area[p1, p2, p3])/(norm[p1 - p2] + norm[p2 - p3] + norm[p3 - p1])
```

l 繪製三角形

```
In[7]:= l[p1_, p2_, p3_] := Graphics[Line[{p1, p2, p3, p1}]];
```

innerpoint 繪製三點的內心

```
In[8]:= innerpoint[p1_, p2_, p3_] := Graphics[{RGBColor[1, 0, 0],
            AbsolutePointSize[4], Point[OI[p1, p2, p3]]}];
```

innercircle 繪製三點的內切圓

```
In[9]:= innercircle[p1_, p2_, p3_] :=
         Graphics[{RGBColor[1, 0, 0], AbsoluteDashing[{4, 4}],
           Circle[OI[p1, p2, p3], r[p1, p2, p3]]}]
```

In[10]:=
```
{p1, p2, p3} = {{0, 0}, {2, 0}, {2, 3}};
Show[l[p1, p2, p3], innerpoint[p1, p2, p3],
 innercircle[p1, p2, p3], AspectRatio → Automatic, Axes → True]
```



Out[11]= - Graphics -

In[68]:=
```
p[p1, p2, p3]
```

Out[68]= - Graphics -

參考資料

1. *Mathematica* 程式設計風格與應用 — 余家銘編著，文魁資訊股份有限公司出版
2. 用 *Mathematica* 學中學數學 — 邱博文著，費因曼文化出版
3. *Mathematica* 5 數學運算大師 — 洪維恩著，旗標出版股份有限公司出版
4. *Mathematica*[數學篇] — 白石修二著，森北出版株式會社出版
5. *Mathematica*[基礎操作篇] — 白石修二著，森北出版株式會社出版
6. 台北市立建國高級中學數學學習資料
7. *Mathematica* 5.0 軟體內建說明